Готовимся к ЕГЭ. Задачи С4

На вход программе подаются произвольно алфавитно-цифровые символы. Ввод этих символов заканчивается точкой.
Требуется написать программу, которая будет печатать последовательность строчных английских букв (“a”, “b”, … ”z”) из входной последовательности и частот их повторения. Печать должна происходить в алфавитном порядке. 
Например, пусть на вход подаются следующие символы:
fhb5kbfыshfm.
В этом случае программа должна вывести
b2
f3
h2
k1
m1
s1

При решении этой задачи нужно воспользоваться тем, что символьный тип данных - это перечислимый тип. Следовательно, символы могут служить индексами элементов массива. Будем вводить входные данные символ за символом, выбирая для анализа только строчные латинские буквы. Для хранения нужных нам данных организуем массив, в котором номерами элементов будут символы, а значениями элементов - количество данных символов во входной строке. По окончанию ввода данных нам останется вывести результат - номера и значения ненулевых элементов нашего массива. 

var
a: array ['a'..'z'] of integer;
n, i, j: integer;
c: char;
begin
assign (input,'input.txt'); reset (input);
assign (output,'output.txt'); rewrite (output);
for c:='a' to 'z' do a[c]:=0;
read (c);
while c <> '.'
do begin
if (c>='a') and (c<='z')
then
a[c]:=a[c]+1;
read (c);
end;
for c:='a' to 'z' do
if a[c]<>0
then writeln (c, a[c]);
close (input);
close (output);
end. 
На вход программе подаются сведения о сдаче экзаменов учениками 9-х классов некоторой средней школы. 
В первой строке сообщается количество учеников N, которое не меньше 10, но не превосходит 100, каждая из следующих N строк имеет следующий формат: <Фамилия> <Имя> <оценки>, где <Фамилия> – строка, состоящая не более чем из 20 символов, <Имя> – строка, состоящая не более чем из 15 символов, <оценки> – через пробел три целых числа, соответствующие оценкам по пятибалльной системе. <Фамилия> и <Имя>, а также <Имя> и <оценки> разделены одним пробелом. Пример входной строки:
Иванов Петр 4 5 3
Требуется написать как можно более эффективную программу (укажите используемую версию языка программирования, например, Borland Pascal 7.0), которая будет выводить на экран фамилии и имена трех худших по среднему баллу учеников. Если среди остальных есть ученики, набравшие тот же средний балл, что и один из трех худших, то следует вывести и их фамилии и имена.

Сложность этой задачи состоит в том, что один и тот же балл могут получить несколько человек. И количество фамилий, которые мы выведем, может значительно различаться при разных входных данных. Обсуждение, посвященное этому смотри в http://community.livejournal.com/iteachers_sar/44470.html 
Я подсмотрела идею решения, выложенную учителем информатики г. Екатеринбурга, на портале «Открытый класс» 
http://www.openclass.ru/node/8188?page=1#comment-7290
. Только немного модернизировала его под Borland Pascal и в свете своего понимания условия. Итак, организуем массив, где номером элемента будет сумма баллов, а типом элемента будет множество номеров фамилий. Возможные суммы баллов колеблются от 0=0+0+0 до 15=5+5+5. Следовательно, массив будет из шестнадцати элементов. Придется организовать еще один массив, в котором фамилии будут соответствовать своим номерам. 
После того, как при считывании входных данных мы запишем данные в подготовленные структуры, нам останется найти непустые элементы-множества. Будем выводить фамилии, чьи номера хранятся в этих множествах, пока количество этих фамилий не станет равным или не превысит 3. 
Получается, что в зависимости от входных данных мы можем вывести или фамилии только из одного множества, или из двух множеств, или из трех множеств. Все зависит от входных данных. Вот. 

var
a: array [0..15] of set of byte;
b: array [1..100] of string[36];
n, m, i, k, c1, c2, c3: byte;
s: string;
c: char;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
readln (n);
for i:=1 to n do b[i]:=''; {массив для хранения фамилий и имен}
for k:=0 to 15 do a[k]:=[]; {массив для хранения соответствия номеров фамилий и баллов}
for i:=1 to n do
begin
s:=''; {считываем фамилию и имя}
m:=0;
read (c);
while m<>2 do
begin
s:=s+c;
read(c);
if c=' ' then m:=m+1;
end;
b[i]:=s; {сохраняем фамилию и имя в массив под очередным номером}
readln (c1, c2, c3); {считываем баллы}
m:=c1+c2+c3; {находим сумму баллов}
a[m]:=a[m]+[i]; {помещаем номер фамилии в элемент массива, чьим номером будет является сумма баллов.}
end;
m:=0;
k:=0;
while (m<3) and (k<=15) do {выводим фамилии и имена тех учащихся, кто получил наименьшую сумму баллов}
begin
if a[k]<>[]
then
for i:=1 to n do
if i in a[k]
then
begin
writeln (b[i]);
m:=m+1;
end;
k:=k+1;
end;
close (input);
close (output);
end.

На вход программе подаются сведения о пассажирах, желающих сдать свой багаж в камеру хранении на заранее известное время до полуночи. В первой строке сообщается количество пассажиров N, которое не меньше 3, но не превосходит 1000, и количество ячеек в камере хранения M, которое не меньше 10, но не превосходит 1000. Каждая из следующих N строк имеет следующий формат: <Фамилия> <время сдачи багажа> <время освобождения ячейки>, где <Фамилия> - строка, состоящая не более чем из 20 символов, <время сдачи багажа> - через двоеточие два целых числа, соответствующие часам (от 00 до 23 – ровно 2 символа) и минутам (от 00 до 59 – ровно 2 символа), <время освобождения ячейки> имеет тот же формат. <Фамилия> и <время сдачи багажа>, <время сдачи багажа> и <время освобождения ячейки> разделены одним пробелом. 
Сведения отсортированы в порядке времени сдачи багажа. Каждому из пассажиров в камере хранения выделяется свободная ячейка с минимальным номером. Если в момент сдачи багажа свободных ячеек нет, то пассажир уходит, не дожидаясь освобождения одной из них.
Требуется написать программу, которая будет выводить на экран для каждого пассажира номер ячейки, которая ему будет предоставлена (можно сразу после ввода данных очередного пассажира). Если ячейка пассажиру не предоставлена, то его фамилия не печатается.
Пример входных данных:
3 10
Иванов 09:45 12:00
Петров 10:00 11:00
Сидоров 12:00 13:12
Результат работы программы на этих входных данных:
Иванов 1
Петров 2
Сидоров 1

В этой задаче самое трудное - организовать ввод данных, перевод их из текстового формата в числовой. В своем решении я храню время освобождения ячейки. В самом условии есть две подсказки (они выделены курсивом). Первая - что все происходит в течение одних суток, а вторая - вывод можно сделать параллельно вводу.

var
a: array [1..1000] of integer; {массив храние времени освобождения ячеек}
n, m, i, k, j, time1, time2, err: integer;
c: char;
s, t, p: string;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
read (n); {количество пассажиров}
readln (m); {количество ячеек}
for i:=1 to m do a[i]:=0;
for i:=1 to n do
begin
s:=''; {фамилия}
read (c);
while c<>' ' do
begin
s:=s+c;
read (c);
end;
read (c);
t:=''; {часы сдачи как символ}
while c<>':' do
begin
t:=t+c;
read (c);
end;
p:=''; {минуты сдачи как символ}
read (c);
while c<>' ' do
begin
p:=p+c;
read (c);
end;
val (t, k, err);
val (p, j, err);
time1:=k*60+j; {время сдачи багажа в минутах}
read (c);
t:=''; {часы освобождения ячейки как символ}
while c<>':' do
begin
t:=t+c;
read (c);
end;
readln (j); {минуты освобождения ячейки}
val (t, k, err);
time2:=k*60+j; {время освобождения ячейки в минутах}
k:=1; {ищем свободную ячеку}
while (k<=m) and (a[k]>time1) do k:=k+1;
if k<=m {если она найдена}
then
begin
a[k]:=time2; {занимаем ее}
writeln (s, k:3); {выводим фамилию пассажира и номер ячейки}
end;
end;
close (input);
close (output);
end.

На вход программе подается последовательность цифр. Ввод этих символов заканчивается точкой (другие символы, отличные от точки и цифр «0» … «9», во входных данных отсутствуют). Требуется написать как можно более эффективную программу, которая будет печатать цифры, встречающиеся во входной последовательности, в порядке уменьшения частоты их встречаемости. Каждая цифра при этом должна быть распечатана один раз.
Если какие-то цифры встречаются одинаковое число раз, то они выводятся по возрастанию. Например, пусть на вход подаются следующие символы:
1232432.
В данном случае программа должна вывести
2314

Но решать будем не ее, а более сложный вариант. 
Вам дана строка, состоящая из строчных букв английского алфавита и заканчивающаяся символом "." (точка).
Требуется написать программу, которая будет выводить буквы, встречающиеся во входной последовательности, в порядке уменьшения частоты их встречаемости.
Формат входных данных
Во входном файле записан текст, состоящий из строчных символов английского алфавита и точки.
Длина текста будет не превышать 100 символов. В тексте встречается хотя бы одна буква.
Формат выходных данных
Выведите все буквы, встречающиеся во входном файле, в порядке уменьшения частоты их встречаемости.
Каждая буква, которая встречается в тексте, должна быть выведена ровно один раз. Точка при этом не учитывается.
Если какие-то буквы встречаются одинаковое количество раз, то они выводятся в алфавитном порядке.
Примеры:
Вход
zzzbbaattt.
Выход
tzab
Откуда появился этот вариант? Такое условие предложили "резиденты" центра олимпиадной подготовки имени Н.Л.Андреевой Саратовского гос. университета. И именно этот вариант реализован в контестере на 
http://school.sgu.ru

Мне мое решение не нравится, оно тяжеловесное, непонятное, но правильное :). Может кто придумает лучше?

type
simbol = record
p: char;
i: integer;
end;
var
a: array ['a'..'z'] of simbol;
k, l, n: integer;
c, h, m: char;
flag: boolean;
tmp: simbol;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
for c:='a' to 'z' do
begin
a[c].p:=c;
a[c].i:=0;
end;
read (c);
while c <> '.' do
begin
a[c].i:=a[c].i+1;
read (c);
end;
flag:=false;
m:='z';
while not flag do
begin
flag:=true;
m:=pred(m);
for c:='a' to m do
if a[c].i < a[succ(c)].i
then
begin
tmp:=a[c];
a[c]:=a[succ(c)];
a[succ(c)]:=tmp;
flag:=false;
end;
end;
c:='a';
while (a[c].i > 0) and (c <= 'z') do
begin
write (a[c].p);
c:=succ(c);
end;
close (input);
close (output);
end.

На вход программе подаются сведения о пассажирах, забронировавших по интернету авиабилеты (только тех, у кого время бронирования еще не истекло). В первой строке задано текущее время: через двоеточие два целых числа, соответствующие часам (от 00 до 23 – ровно 2 символа) и минутам (от 00 до 59 – ровно 2 символа). Во второй строке сообщается количество пассажиров N, которое не меньше 3, но не превосходит 1000. Каждая из следующих N строк имеет следующий формат: <Фамилия> <время окончания брони>, где <Фамилия> - строка, состоящая не более чем из 20 символов, <время окончания брони> - через двоеточие два целых числа, соответствующие часам (от 00 до 23 – ровно 2 символа) и минутам (от 00 до 59 – ровно 2 символа). <Фамилия> и <время окончания брони> разделены одним пробелом. Сведения отсортированы в порядке времени, когда производилось бронирование. Все значения времени относятся к текущим суткам.
Требуется написать эффективную программу, которая в хронологическом порядке (то есть в порядке возрастания значения времени окончания брони) выедет фамилии пассажиров, у которых в ближайшие 3 часа текущего дня закончится броня.
Пример входных данных:
10:00
3
Иванов 13:00
Петров 10:00
Сидоров 13:12
Результаты работу программы для этого примера:
Петров
Иванов

Я специально в тексте задачи выделила курсивом фразу о том, что все указанные времена относятся к одним суткам. Следовательно, нам не надо решать проблему смены даты (что в этой задаче было бы самым сложным, но не катастрофичным). Еще одна проблема - это хранение и сравнение времен. Я, в своем решении, все времена переводила в минуты.Итак решение. 

type
pass = record
s: string[20];
t: integer;
end; 
var
a: array [1..1000] of pass; {массив для хранения фамилий пассажиров и времен окончания брони}
tmp: pass;
n, m, i, k, j, time1, time2, err: integer;
c: char;
s, t: string;
flag: boolean;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
read (c);
t:=''; {текущий час в строковом формате}
while c<>':' do
begin
t:=t+c;
read (c);
end;
readln (j); {текущие минуты}
val (t, k, err);
time1:=k*60+j; {текущее время в числовом формате}
readln (n); {количество пассажиров}
m:=1; {количество истекающих броней}
for i:=1 to n do
begin
s:=''; {фамилия пассажира}
read (c);
while c<>' ' do
begin
s:=s+c;
read (c);
end;
read (c);
t:=''; {часы окончания брони в строковом формате}
while c<>':' do
begin
t:=t+c;
read (c);
end;
readln (j); {минуты окончания брони}
val (t, k, err);
time2:=k*60+j; {время окончания брони в числовом формате}
if time2-time1<=180 {если время истечения брони меньше или равно трем часам (180 минутам)}
then
begin
a[m].s:=s; {то запоминаем фамилию пассажира}
a[m].t:=time2; {и время истечения брони}
m:=m+1;
end;
end;
j:=m-1; {сортировка созданного массива методом "пузырька" по полю времени окончания брони}
flag:=false;
while not (flag) do
begin
flag:=true;
j:=j-1;
for i:=1 to j do
if a[i].t>a[i+1].t
then
begin
tmp:=a[i];
a[i]:=a[i+1];
a[i+1]:=tmp;
flag:=false;
end;
end;
for i:=1 to m-1 do {вывод результата - фамилий в отсортированном по времени порядке}
writeln (a[i].s);
close (input);
close (output);
end.
На вход программе подаются сведения обо всех учениках некоторой средней школы. В первой строке сообщается количество учеников N, каждая из следующих N строк имеет следующий формат: <Фамилия> <Имя> <класс>, где <Фамилия> - строка, состоящая не более чем из 20 символов, <Имя> - строка, состоящая не более чем из 15 символов, <класс> - год обучения (от 1 до 12) и латинская заглавная буква (от «A» до «Z») без пробела. <Фамилия> и <Имя>, а также <Имя> и <класс> разделены одним пробелом. Пример входной строки:
Иванов Петр 10А
В рамках одной параллели классы нумеруются буквами подряд, начиная с «А». В некоторых параллелях классов может не быть совсем.
Требуется написать как можно более эффективную программу, которая будет выводить на экран информацию о параллелях (годе обучения) с наибольшем количеством классов. Программа должна выводить на экран в первой строке количество классов в искомых параллелях, а во второй строке – номера этих параллелей в порядке возрастания через пробел. Например:
3
1 7 11

Решение на Паскале

var
a: array [1..12] of char;
i, j, n, m, err: integer; k: char;
s, t: string;
begin
assign (input, 'input.txt');
assign (output, 'output.txt');
reset (input);
rewrite (output);
readln (n);
for i:=1 to 12 do a[i]:=' '; {В этом массиве мы будем хранить последнюю букву для каждой параллели}
for i:=1 to n do begin
readln (s);
j:=length (s); {Так как номер параллели и буква - это последние символы, то нам не интересно все остальное}
k:=s[j]; {Это буква класса}
j:=j-1;
t:='';
while s[j]<>' '
do begin
t:=s[j]+t;
j:=j-1;
end;
val (t, m, err); {Номер параллели}
if a[m] < k then a[m]:=k; {Если встретилась более "старшая" буква, записываем ее в массив в элемент, соответствующий параллели}
end;
k:=a[1]; {Ищем "максимальную" букву}
for i:=1 to 12 do
if a[i] > k
then
k:=a[i];
m:=ord(k)-ord('A')+1; {Находим количество классов в самой длинной параллели}
writeln (m); 
for i:=1 to 12 do
if a[i]=k then write (i:3); {Выводим номера этих параллелей}
close (input);
close (output);
end.

Я нарочно подробно не описала решение. Хочу вопросов!

Я хочу начать разбирать задачи С4 - самые трудные в ЕГЭ по информатике. Каждую неделю буду выкладывать одну задачу с краткими комментариями. Надеюсь на вопросы - умные, глупые и т.д. Самое главное, чтобы они были. :)
Эти же задачи будут разбираться и на 
http://school.sgu.ru
. Вопросы можно будет задавать и там (только для зарегистрированных пользователей). Если Вы хотите использовать эти решения для своих статей, портфолио и т.д. - ссылка обязательна! А то мне очень обидно. ;)
Итак...

На вход программе подаются сведения о телефонах всех сотрудников некоторого учреждения. В первой строке сообщается количество сотрудников N, каждая из следующих N строк имеет следующий формат: <Фамилия> <Инициалы> <телефон>, где <Фамилия> - строка, состоящая не более чем из 20 символов, <Инициалы> - строка, состоящая не более чем из 4-х символов (буква, точка, буква, точка), <телефон> - семизначный номер, 3-я и 4, я, а также 5-я и 6-я цифры которого разделены символом «-». <Фамилия> и <Инициалы>, а также <Инициалы> и <телефон> разделены одним пробелом. Пример входной строки:

Иванов П.С. 555-66-77

Сотрудники одного подразделения имеют один и тот же номер телефона. Номера телефонов в учреждении отличаются только двумя последними цифрами.

Требуется написать как можно более эффективную программу, которая будет выводить на экран информацию, сколько в среднем сотрудников работает в одном подразделении данного учреждения.



Теперь решение на Паскале:

var
c: set of byte;
n, i, k, j, l: byte;
b: char;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
readln (n); {количество сотрудников}
k:=0; {счетчик подразделений}
c:=[ ]; {множество для хранения номеров подразделений}
for i:=1 to n do
begin
j:=0;
while j<>2 do
begin
read (b);
if b='-' then j:=j+1;
end;
readln (l); {считываем номер подразделения}
if not (l in c) {если он еще не встречался}
then
begin
k:=k+1; {прибавляем единицу к счетчику}
c:=c+[l]; {помещаем номер подразделения во множество}
end;
end;
writeln (n/k:5:3); {находим среднее количество сотрудников в подразделении}
close (input);
close (output);
end.


Сразу хочу отметить, что это идею решения предложил одним из моих учеников, за что ему отдельное Браво! Итоговая реализация моя. 

Небольшой комментарий. Здесь не нужно хранить фамилии и имена, не нужно хранить полные номера телефонов. Следовательно, нам интересны только последние две цифры телефона, которые мы отсекаем и анализируем. Всего различных двузначных чисел 100 штук - от 00 до 99. Следовательно для их хранения нам можно завести множество.

Смотрим на последние две цифры телефона. Если они уже встречались, мы просто пропускаем их. Если телефон до этого не встречался, то мы помещаем его в множество и прибавляем единицу к счетчику подразделений.

После обработки списка делим количество всех сотрудников на количество подразделений.
И новая задача - 
На вход программе подаются сведения о ячейках автоматической камеры хранения багажа. В первой строке задана текущая дата: через точку два целых числа, соответствующие дню (от 01 до 31 – ровно 2 символа) и месяцу (от 01 до 12 – ровно 2 символа). Во второй строке сообщается количество занятых ячеек N, которое не меньше 3, но не превосходит 1000. Каждая из следующих N строк имеет следующий формат: <номер ячейки> <дата сдачи багажа>, где <номер ячейки> – четырехзначное число, <дата сдачи багажа> – через точку два целых числа, соответствующие дню (от 01 до 31 – ровно 2 символа) и месяцу (от 01 до 12 – ровно 2 символа). Номер ячейки и дата сдачи багажа разделены одним пробелом. Сведения отсортированы в порядке номеров ячеек.
Время хранения багажа – трое суток. Требуется написать как можно более эффективную программу, которая выведет номера ячеек, в которых багаж хранится заведомо больше трех суток – то есть разница между датой сдачи багажа и текущей датой составляет 4 и более дней. Номера ячеек следует выводить в хронологическом порядке сдачи багажа.
Багаж мог сдаваться только в текущем или предыдущем месяце текущего календарного года (если текущий месяц январь, то данные о сдаче багажа в декабре прошлого года отсутствуют). Количество дней в каждом из месяцев текущего года следующее: январь – 31, февраль – 28, март – 31, апрель – 30, май – 31, июнь – 30, июль – 31, август – 31, сентябрь – 30, октябрь – 31, ноябрь – 30, декабрь – 31. Все входные данные корректны.
Пример входных данных:
04.06
3
1000 01.06
1001 31.05
2007 21.05
Результат работы программы для этого примера
2007
1001 
Если в ранее разобранных задачах о багаже или о бронировании авиабилетов не было перехода "через месяц" ("через сутки"), то данная задача как раз и направлена на проверку понимания такого перехода. 
type
bag = record {запись для хранения информации о просроченном багаже}
num: integer; {номер ячейки}
day: integer; {количество дней хранения}
end;
var
a: array [1..1000] of bag; {массив для хранения информации о простроченном багаже}
i, k, n, m, err, edge, p: integer;
data, mes, cur_data,cur_mes: integer;
c: char;
s: string;
tmp: bag;
flag: boolean;
begin
assign (input, 'input.txt'); reset (input);
assign (output, 'output.txt'); rewrite (output);
read (c);
s:=c;
read (c);
s:=s+c;
val (s, cur_data, err); {текущая дата}
read (c);
read (c);
s:=c;
read (c);
s:=s+c;
val (s, cur_mes, err); {текущий месяц}
readln (n); {количество занятых ячеек}
m:=0; {счетчик просроченного багажа}
for i:=1 to n do
begin
read (p);
read (c);
read (c);
s:=c;
read (c);
s:=s+c;
val (s, data, err); {дата сдачи багажа}
read(c);
read (c);
s:=c;
read (c);
s:=s+c;
val (s, mes, err); {месяц сдачи багажа}
if cur_mes=mes {Если багаж сдан в этом месяце}
then
begin
if cur_data-data > 3 {и количество дней хранения превышает 3}
then
begin {сохраняем информацию о просроченном багаже в массив}
m:=m+1;
a[m].num:=p;
a[m].day:=cur_data-data;
end
end
else {Если багаж сдан в прошлом месяце}
begin
case mes of {находим количество дней в прошлом месяце}
1, 3, 5, 7, 8, 10, 12: edge:=31;
2: edge:=28;
4, 6, 9, 11: edge:=30;
end;
if edge-data+cur_data > 3 {Если количество дней хранения превышает 3}
then
begin {сохраняем информацию о просроченном багаже в массив}
m:=m+1;
a[m].num:=p;
a[m].day:=edge+cur_data-data;
end
end;
end;
flag:=true; {Сортируем массив с информацией о просроченном багаже по полю - количество дней хранения}
k:=m;
while flag do
begin
flag:=false;
k:=k-1;
for i:=1 to k do
if a[k].day < a[k+1].day
then
begin
tmp:=a[k];
a[k]:=a[k+1];
a[k+1]:=tmp;
flag:=true;
end;
end;
for i:=1 to m do
writeln (a[i].num); {Вывод результатов}
close (input);
close (output); end.
